CHAOTIC CIRCUIT BEHAVIOR

Many physical systems exhibit steady-state behavior that is oscillatory but not periodic. Until recently, such behavior was thought to be due to some inherent source of randomness in the system and was classified as "noise." Chaos refers to non-periodic asymptotic behavior in systems that are *completely deterministic.* This article describes a number of simple deterministic electronic circuits that exhibit chaos.

Since the pioneering days of electronics in the 1920s, dc equilibrium, periodic, and quasi-periodic steady-state solutions of electronic circuits have been correctly identified and classified. By contrast, the existence of chaos has been widely acknowledged only in the past 30 years.

Even though the notion of chaotic behavior in dynamical systems has existed in the mathematics literature since Poincaré's work at the turn of the century, unusual behavior in the physical sciences as recently as the 1970s was being described as "strange" (1). Today we classify as *chaos* bounded recurrent motion in a deterministic dynamical system that is

J. Webster (ed.), *Wiley Encyclopedia of Electrical and Electronics Engineering.* Copyright © 1999 John Wiley & Sons, Inc.

characterized by sensitive dependence on initial conditions (2,3).

Although the future behavior of a chaotic system is in principle determined exactly by the initial conditions, sensitive dependence on initial conditions means that the precision with which these conditions must be specified grows exponentially with the length of the prediction. Thus a real chaotic system appears to exhibit "randomness" in the time domain because its initial conditions cannot be specified with sufficient precision to make accurate long-term predictions of its behavior.

The earliest experimental observations of chaos in electronic circuits were in forced *nonautonomous* nonlinear oscillators, including the sinusoidally excited neon bulb relaxation oscillator studied by van der Pol and van der Mark (4,5), the forced negative-resistance oscillator of Ueda (6) and the driven series-tuned *RL*-diode circuit (7–9). More recently chaos has been observed and studied in a variety of unforced *autonomous* electronic circuits such as Chua's oscillator (10–12), hysteresis oscillators (13–15), classical circuits such as the Colpitts oscillator (16,17) and the phase-locked loop (18), and a number of important discrete-time systems, including switched capacitor circuits (19), dc–dc converters (20,21), digital filters (22), and sigma-delta modulators (23).

In the following sections we discuss a number of autonomous and nonautonomous chaotic circuits. While exhibiting a rich variety of complex dynamical behaviors, these circuits are simple enough to be constructed and modeled using standard electronic parts and simulators.

CHAOTIC CIRCUITS

Chaos may be defined as bounded steady-state behavior in a deterministic dynamical system that is not an equilibrium point, not periodic, and not quasi-periodic (24).

Solutions of a dissipative deterministic dynamical system eventually settle into regions of the state space called attracting limit sets or attractors. Electronic circuits are typically dissipative due to resistive heating losses; consequently, their long-term behavior is usually characterized by motion on attractors. Here, we consider only dissipative circuits.

Attracting equilibrium point, periodic, and quasi-periodic solutions of deterministic dynamical systems have the property that trajectories from nearby initial conditions that converge to the same limit set become correlated with time. By contrast, two trajectories started close together on an attracting *chaotic* limit set diverge exponentially and soon become uncorrelated; this is called *sensitive dependence on initial conditions* and gives rise to long-term unpredictability.

Technically a chaotic circuit is one whose steady-state behavior is characterized by one or more positive Lyapunov exponents. Lyapunov exponents characterize the average exponential rate of separation of trajectories of a dynamical system on the attractor. Negative Lyapunov exponents cause trajectories to converge with time. If an attractor has a positive Lyapunov exponent, then nearby trajectories on the attractor are separated, on average, along some direction. In practical terms, this means that trajectories of the circuit are unstable yet bounded. Instability means that nearby trajectories diverge on average, and boundedness implies that they remain in some finite volume of the state space.

How can nearby trajectories diverge exponentially and yet remain within a bounded limit let? This may be achieved by repeated *stretching* and *folding* of the flow, as shown in Fig. 1.

Consider the spiral attractor shown in Fig. 1. A trajectory spirals away from the equilibrium point P_- along the plane $E^c(P_-)$ until it enters the D_0 region, where it is folded back into D_{-1} and returns to the plane $E^c(P_-)$ close to P_-. The recurrent stretching and folding continues ad infinitum, producing a chaotic steady-state solution (12).

Note that two trajectories passing very close to $\mathbf{X}_0$ on $E^c(P_-)$ are separated quite dramatically when they cross the plane U_{-1} and enter D_0. By the time they return to D_{-1}, they are no longer close. This illustrates sensitive dependence on initial conditions.



Figure 1. Stretching and folding mechanism of chaos generation in Chua's oscillator. (a) Simulated spiral chaotic attractor showing affine regions (D_{-1} and D_1), separating planes (U_{-1} and U_1), equilibrium points (P_-, 0, and P_+), and their associated eigenspaces (E^r and E^c). (b) Experimentally observed attractor. Vertical axis: V_1 (1 V/div); horizontal axis: V_2 (200 mV/div). Positive-going intersections of the trajectory through the plane defined by $I_3 = 1.37$ mA are shown highlighted.



Figure 2. Experimental manifestations of chaos in the double-scroll attractor from Chua's oscillator (R = 1800 Ω, C_1 = 9.4 nF) (a) Two-dimensional projection of the attractor in state space; vertical axis: V_1 (1 V/div); horizontal axis: V_2 (200 mV/div). (b) Time-domain waveforms. Upper trace: $V_1(t)$ (2 V/div); lower trace: $V_2(t)$ (500 mV/div); horizontal axis: t (2 ms/div). (c) Power spectrum of $V_2(t)$. Vertical axis: power (dB); horizontal axis: frequency (kHz). (d) Time-domain waveforms showing sensitivity to initial conditions. Vertical axis: $V_1(t)$ (2 V/div); horizontal axis: t (500 μs/div).

Chaos is characterized by repeated *stretching and folding* of a bundle of trajectories in state space. In the time domain a chaotic trajectory is neither periodic nor quasi-periodic but looks unpredictable in the long term. This long-term unpredictability manifests itself in the frequency domain as a broad "noiselike" power spectrum (25).

Figure 2 shows experimental manifestations of chaos in the well-known double-scroll chaotic attractor from Chua's oscillator (26).

CHAOS IN AUTONOMOUS ELECTRONIC CIRCUITS

In order to exhibit chaos, an autonomous circuit consisting of resistors, capacitors, and inductors must contain (1) at least one active resistor, (2) at least one nonlinear element, and (3) at least *three* energy-storage elements. The active resistor supplies energy to separate trajectories, the nonlinearity provides folding, and the three-dimensional state space permits persistent stretching and folding in a bounded region without violating the noncrossing property of trajectories.

In this section we consider three important classes of autonomous electronic circuits: Chua's oscillator, Saito's hysteresis oscillator, and the Colpitts oscillator.

Chua's Oscillator

Chua's oscillator (shown in Fig. 3) consists of a linear inductor, two linear resistors, two linear capacitors, and a single voltage-controlled nonlinear resistor N_R, called a Chua diode (10–12). N_R is a voltage-controlled piecewise-linear resistor



Figure 3. Chua's oscillator.



Figure 4. The V–I characteristic of the nonlinear resistor N_R in Chua's oscillator has breakpoints at $\pm E$ and slopes G_a and G_b in the inner and outer regions.

(27) whose continuous odd-symmetric three-segment V–I characteristic (shown in Fig. 4) is defined explicitly by the relationship

$$I_R = G_b V_R + \tfrac{1}{2}(G_a - G_b)(|V_R + E| - |V_R - E|)$$

Chua's oscillator is described by three ordinary differential equations:

$$\begin{aligned}\frac{dV_1}{dt} &= \frac{G}{C_1}(V_2 - V_1) - \frac{1}{C_1}f(V_1)\\ \frac{dV_2}{dt} &= \frac{G}{C_2}(V_1 - V_2) + \frac{1}{C_2}I_3\\ \frac{dI_3}{dt} &= -\frac{1}{L}V_2 - \frac{R_0}{L}I_3\end{aligned}$$

where $G = 1/R$ and $f(V_R) = G_b V_R + \frac{1}{2}(G_a - G_b)(|V_R + E| - |V_R - E|)$.

Chua's circuit is a special case of Chua's oscillator where $R_0 \equiv 0$ (11,12). In practice, an inductor typically has a nonzero series parasitic resistance, implying that $R_0 > 0$. Therefore we consider only the general case of Chua's oscillator.

The primary motivation for studying Chua's oscillator is that it can exhibit *every dynamical behavior* known to be possible in an autonomous three-dimensional continuous-time dynamical system described by a continuous odd-symmetric three-region piecewise-linear vector field. In particular, it can exhibit equilibrium point, periodic, quasi-periodic, and chaotic steady-state solutions. The oscillator is also useful in studying *bifurcations* and *routes to chaos.* A user-friendly program for studying chaos in Chua's circuit is available (28).

A bifurcation is a qualitative change in the behavior of a system (2). One of the most familiar bifurcations in electronic circuits is the *Hopf bifurcation,* where a circuit that had been at an equilibrium point begins to oscillate when a parameter is increased through some critical value called a bifurcation point.

A well-defined sequence of bifurcations that takes a system from dc or periodic behavior to chaos is called a *route to chaos.* With appropriate choices of its component values, Chua's oscillator can follow the *period-doubling, intermittency,* or *quasi-periodic* route to chaos.

Example: Period-Doubling Route to Chaos in Chua's Oscillator. The period-doubling route to chaos is characterized by a cascade of period-doubling bifurcations. Each period-doubling transforms a limit cycle into one at half the frequency, spreading the energy of the system over a wider range of frequencies. An infinite cascade of such doublings results in a chaotic trajectory of infinite period and a broad frequency spectrum that contains energy at all frequencies. Figure 5 is a set of snapshots of the period-doubling route to chaos in Chua's oscillator.

Bifurcation Diagrams

While state-space, time-, and frequency-domain measurements are useful for characterizing steady-state behaviors, nonlinear dynamics offers several other tools for summarizing qualitative information concerning bifurcations.

A *bifurcation diagram* is a plot of the attractors of a system versus a control parameter. For each value of the control parameter, called the *bifurcation parameter,* one plots samples of a state of the system. In the case of a fixed point, all samples are identical and the attractor appears on the bifurcation diagram as a single point. If a periodic solution is sampled synchronously, the attractor appears in the bifurcation diagram as a finite set of points. A periodic solution consisting of



Figure 5. Period-doubling in Chua's oscillator with L = 18 mH, R_0 = 12.5 Ω, C_2 = 100 nF, G_a = −50/66 mS = −757.576 μS, G_b = −9/22 mS = −409.091 μS, and E = 1 V. Simulated state space trajectories (left), time waveforms $V_1(t)$ (top right), and power spectra of $V_2(t)$ (bottom right). (a) G = 530 μS: periodic steady state; (b) G = 537 μS: period two; (c) G = 539 μS: period four; (d) G = 541 μS: spiral chaotic attractor.



Figure 6. Bifurcation diagram for V_1 in Chua's oscillator.

n points is called a "period-n" orbit. Since a chaotic solution is nonperiodic, sampling produces an uncountable set of points.

When producing a bifurcation diagram, the sampling instants are determined by a clock that is derived from the dynamics of the system under consideration. In discrete systems, one simply plots successive values of a state variable. For nonautonomous continuous-time systems with periodic forcing, the driving signal provides a natural sampling clock. Some type of discretization in time is needed for autonomous continuous-time systems. In this case, the sampling instants are defined by crossings of a trajectory of the system through a reference plane in the state space that is called a *Poincaré section*.

Figure 6 shows a simulated bifurcation diagram for V_1 in Chua's oscillator as the bifurcation parameter G is swept from 533 to 543 μS. V_1 is sampled when $V_2 = 0$. Period-one, period-two, and period-four orbits for $G = 530$, 537, and 539 μS yield one, two, and four points, respectively, on the bifurcation diagram.

Chaos Generation Mechanism in Chua's Oscillator

Because of the piecewise-linear nature of the nonlinearity $f(\cdot)$, the vector field of Chua's oscillator may be decomposed into three distinct affine regions—$V_1 < -E$, $|V_1| \leq E$, and $V_1 > E$—which are called the D_{-1}, D_0, and D_1 regions, respectively (12). In each region, the dynamics are linear. The global dynamics may be determined by considering separately the behavior in each of the three regions (D_{-1}, D_0, and D_1) and then gluing the pieces together along the boundary planes U_{-1} and U_1.

Shil'nikov Chaos in Chua's Oscillator

In the following discussion, consider a fixed set of component values: $L = 18$ mH, $R_0 = 12.5\ \Omega$, $C_2 = 100$ nF, $C_1 = 10$ nF, $G_a = -50/66$ mS $= -757.576\ \mu$S, $G_b = -9/22$ mS $= -409.091\ \mu$S, and $E = 1$ V. When $G = 550\ \mu$S, the oscillator has three equilibrium points at P_+, 0, and P_-. The equilibrium point at the origin (0) has one real eigenvalue γ_0 and a complex conjugate pair $\sigma_0 \pm j\omega_0$. The outer equilibria (P_- and P_+) each have a real eigenvalue γ_1 and a complex conjugate pair $\sigma_1 \pm j\omega_1$.

Dynamics of D_0

A trajectory starting from some initial state in the D_0 region may be decomposed into its components along the plane $E^c(0)$ and the vector $E^r(0)$. When $\gamma_0 > 0$ and $\sigma_0 < 0$, the component along $E^c(0)$ spirals toward the origin along this plane, while the component in the direction $E^r(0)$ grows exponentially. Adding the two components, we see that a trajectory starting slightly above the plane $E^c(0)$ spirals toward the origin along the $E^c(0)$ direction, all the while being pushed away from $E^c(0)$ along the unstable direction $E^r(0)$. As the (stable) component along $E^c(0)$ shrinks in magnitude, the (unstable) component grows exponentially.

Thus the trajectory follows a helix of exponentially decreasing radius whose axis lies in the direction of $E^r(0)$; this is illustrated in Fig. 7.

Dynamics of D_{-1} and D_1

A trajectory starting from some initial state in the D_1 region may be decomposed into its components along the plane $E^c(P_+)$ and the vector $E^r(P_+)$. When $\gamma_1 < 0$ and $\sigma_1 > 0$, the component on $E^c(P_+)$ spirals away from P_+ along this plane, while the component in the direction of $E^r(0)$ tends asymptotically toward P_+. Adding the two components, we see that a trajectory starting close to the real eigenvector $E^r(P+)$ above the plane moves toward $E^c(P_+)$ along a helix of exponentially increasing radius. Since the component along $E^r(P_+)$ shrinks exponentially in magnitude and the component on $E^c(P_+)$ grows exponentially, the trajectory is quickly flattened onto $E^c(P_+)$, where it spirals away from P_+ along the plane; this is illustrated in Fig. 8.

By symmetry, the equilibrium point P_- in the D_{-1} region has three eigenvalues: γ_1 and $\sigma_1 \pm j\omega_1$. The vector $E^r(P_-)$ is associated with the real eigenvalue γ_1; the real and imaginary parts of the eigenvectors associated with the complex conjugate pair $\sigma_1 \pm j\omega_1$ define a plane $E^c(P_-)$ along which trajectories spiral away from P_-.

Global Dynamics

With the given set of parameter values, the equilibrium point at the origin has an *unstable* real eigenvalue and a *stable* pair



Figure 7. Dynamics of the D_0 region.



Figure 8. Dynamics of the D_1 region. By symmetry, the D_{-1} region has equivalent dynamics.



Figure 9. A simulated double-scroll attractor in Chua's oscillator with $G = 565\ \mu$S.

of complex conjugate eigenvalues; the outer equilibrium point P_- has a *stable* real eigenvalue and an *unstable* complex pair.

In particular, P_- has a pair of unstable complex conjugate eigenvalues $\sigma_1 \pm \omega_1$ ($\sigma_1 > 0$, $\omega_1 \neq 0$) and a stable real eigenvalue γ_1, where $|\sigma_1| < |\omega_1|$. One can prove that the circuit is *chaotic in the sense of Shil'nikov* by showing, in addition, that it possesses a homoclinic orbit for this set of parameter values. A homoclinic orbit is a closed trajectory that is asymptotic in forward and reverse time to the same equilibrium point. Trajectories that lie close to a homoclinic orbit exhibit complex dynamics.

A trajectory starting on the vector $E^r(0)$ close to 0 moves away from the equilibrium point until it crosses the boundary U_1 and enters D_1. If this trajectory is *folded* back into D_0 by the dynamics of the outer region, and reinjected toward 0 along the stable plane $E^c(0)$, then the required *homoclinic orbit* is produced. That Chua's oscillator is chaotic in the sense of Shil'nikov was first proved by Chua et al. in 1985 (26).

Double-Scroll Attractor

The double-scroll attractor, a two-dimensional projection of which is shown in Fig. 9, is a chaotic attractor in Chua's oscillator. This strange attractor is so called because of the intertwined scroll-like structure of a transverse section through the attractor at the origin.

Practical Implementation of Chua's Oscillator

Chua's oscillator can be realized in a variety of ways using standard or custom-made electronic components. All of the linear elements (capacitor, resistor, and inductor) are readily available as two-terminal devices. A nonlinear resistor N_R with the prescribed V–I characteristic (called a Chua diode) can be implemented by connecting two negative-resistance converters in parallel, as shown in Fig. 10 (29). A complete list of components for this circuit is given in Table 1. Chua diodes have also been implemented in integrated circuit form (30).

The op amp subcircuit consisting of A_1, A_2 and R_1–R_6 functions as a negative-resistance converter N_R with a V–I charac-



Figure 10. Practical implementation of Chua's oscillator using two op amps and six resistors to realize the Chua diode.

Table 1. Component List for the Practical Implementation of Chua's Oscillator Shown in Fig. 10

Element	Description	Value
A_1	Op amp (1/2 AD712 or equivalent)	
A_2	Op amp (1/2 AD712 or equivalent)	
C_1	Capacitor	10 nF
C_2	Capacitor	100 nF
R	Potentiometer	2 kΩ
R_1	1/4 W resistor	3.3 kΩ
R_2	1/4 W resistor	22 kΩ
R_3	1/4 W resistor	22 kΩ
R_4	1/4 W resistor	2.2 kΩ
R_5	1/4 W resistor	220 Ω
R_6	1/4 W resistor	220 Ω
L, R_0	Inductor (TOKO type 10RB)	18 mH, 12.5 Ω



Figure 11. Every physically realizable nonlinear resistor N_R is eventually passive—the outermost segments must lie completely within the first and third quadrants of the V_R–I_R plane for sufficiently large $|V_R|$ and $|I_R|$.

teristic as shown in Fig. 11. Using two 9 V batteries to power the op amps gives $V^+ = 9\ V$ and $V^- = -9\ V$. From measurements of the saturation levels of the AD712 outputs, $E_{sat} \approx 8.3\ V$, giving $E \approx 1\ V$. With $R_2 = R_3$ and $R_5 = R_6$, the nonlinear characteristic is defined by $G_a = -1/R_1 - 1/R_4 = -50/66$ mS, $G_b = 1/R_3 - 1/R_4 = -9/22$ mS, and $E = R_1E_{sat}/(R_1 + R_2) \approx 1$ V. Note that the real inductor is modeled as a series connection of an ideal linear inductor L and a linear resistor R_0.

Nonideality of an Op amp–Based Chua Diode

The V–I characteristic of the op amp–based Chua diode in Fig. 10 differs from the desired piecewise-linear characteristic shown in Fig. 4 in that it has *five* segments, the outer two of which have positive slopes $G_c = 1/R_5 = 1/220$ S.

This nonideality is due to the fundamental laws of nature. Any physical realization of a nonlinear resistor is *eventually passive,* meaning simply that for a large enough voltage across its terminals, the instantaneous power $P_R(t) = (V_R(t)I_R(t))$ consumed by the device is positive.

Hence the V–I characteristic of a real Chua dioide must include at least two outer segments with positive slopes that return the characteristic to the first and third quadrants. From a practical point of view, as long as the voltages and currents on the attractor are restricted to the negative-resistance region of the characteristic, these outer segments will not affect the circuit's behavior.

SPICE Simulation of Chua's Oscillator

Chaotic circuits may be readily simulated using commercial circuit simulators such as SPICE (31). Figure 12 shows a net-list for the practical implementation of Chua's oscillator shown in Fig. 10. The AD712 op amps in this realization of the circuit are modeled using Analog Devices' AD712 macromodel (32). The TOKO 10 RB inductor has a nonzero series resistance, which we have included in the SPICE net-list: `R0` = 12.5 Ω. Node numbers are as in Fig. 10: The power rails are 111 and 222; 10 is the "internal" node of our physical inductor where its series inductance is connected to its series resistance.

A double-scroll attractor results from our SPICE 3e2 simulation using the input deck shown in Fig. 12; this attractor is plotted in Fig. 13.

```
CHUA'S OSCILLATOR

L    1  10   0.018
R0   10 0    12.5
R    1  2    1770
C2   1  0    100.0N
C1   2  0    10.0N
* 2-VNIC CHUA DIODE
V+  111  0    DC 9
V-  0   222  DC 9
XA1  2  4 111 222 3 AD712
R1   4  0    3.3K
R2   3  4    22K
R3   2  3    22K
XA2  2  6 111 222 5 AD712
R4   6  0    2.2K
R5   5  6    220
R6   2  5    220

* AD712 SPICE Macro-model   1/91, Rev. A
* Copyright 1991 by Analog Devices, Inc.
* (reproduced with permission)
*
.SUBCKT AD712 13 15 12 16 14
*
VOS 15 8 DC 0
EC 9 0 14 0 1
C1 6 7 .5P
RP 16 12 12K
GB 11 0 3 0 1.67K
RD1 6 16 16K
RD2 7 16 16K
ISS 12 1 DC 100U
CCI 3 11 150P
GCM 0 3  0 1  1.76N
GA 3 0  7 6  2.3M
RE 1 0 2.5MEG
RGM 3 0 1.69K
VC 12 2 DC 2.8
VE 10 16 DC 2.8
RO1 11 14 25
CE 1 0 2P
RO2 0 11 30
RS1 1 4 5.77K
RS2 1 5 5.77K
J1 6 13 4 FET
J3 7 8 5 FET
DC 14 2 DIODE
DE 10 14 DIODE
DP 16 12 DIODE
D1 9 11 DIODE
D2 11 9 DIODE
IOS 15 13 5E-12
.MODEL DIODE D
.MODEL FET PJF(VTO=-1 BETA=1M IS=25E-12)
.ENDS

.IC V(1)=0 V(2)=0.1
.TRAN 0.01MS 100MS 50MS
.OPTIONS RELTOL=1.0E-5 ABSTOL=1.0E-5
.PRINT TRAN V(1) V(2)
.END
```

Figure 12. SPICE deck to simulate the transient response of the implementation of Chua's oscillator in Fig. 10. The op amps are modeled by the Analog Devices AD712 macromodel. `R0` models the series resistance of the real inductor L.



Figure 13. SPICE simulation of a double-scroll attractor in Chua's oscillator.

Hysteretic Chaotic Oscillator

A hysteretic chaotic oscillator is one in which the nonlinear elements exhibits "hysteretic" behavior resulting from slow-fast dynamics (33).

"Hysteretic" behavior in electronic circuits, such as that which occurs in a Schmitt trigger or a nonmonotone nonlinear resistor, is normally associated with fold bifurcations; it manifests itself as "jumps" in voltages or currents at impasse points (13,27,34).

The fast dynamics associated with a nonmonotone current-controlled (voltage-controlled) negative resistor can be modeled by a small transit inductance (capacitance) in series (parallel) with the resistor (35).

Saito's oscillator, shown in Fig. 14, contains a nonmonotone current-controlled "hysteresis" resistor N_R (14,15). L_0 is a small transit inductance that completes the model.

This circuit is described by a system of three autonomous state equations:

$$
\begin{aligned}
\frac{dV_1}{dt} &= \frac{1}{C}I_2 - \frac{1}{C}I_3 \\
\frac{dI_2}{dt} &= -\frac{1}{L}V_1 - \frac{R}{L}I_2 \\
\frac{dI_3}{dt} &= \frac{1}{L_0}V_1 - \frac{1}{L_0}g(I_3)
\end{aligned}
$$

where $g(I_R) = R_b I_R + \frac{1}{2}(R_a - R_b)(|I_R + I| - |I_R - I|)$.



Figure 14. Saito's oscillator.



Figure 15. The V–I characteristic of the nonlinear resistor N_R in Saito's oscillator has breakpoints at $\pm I$ and slopes R_a and R_b in the inner and outer regions.

In the limit as $L_0 \to 0$, the third equation imposes the constraint

$$V_1 = g(I_3)$$

Trajectories are thus constrained to lie along the driving-point characteristic of the nonlinear resistor N_R. On the outer segments of this characteristic,

$$I_3 = V_1 \pm E_s$$

where the intercepts $\pm E_s$ are as shown in Fig. 15. In these regions, the system is governed by two-dimensional dynamics

$$
\begin{aligned}
\frac{dV_1}{dt} &= \frac{1}{C}I_2 - \frac{1}{RC}(V_1 \pm E_s) \\
\frac{dI_2}{dt} &= -\frac{1}{L}V_1 - \frac{R}{L}I_2
\end{aligned}
$$

If the trajectory is on the upper segment of the V–I characteristic and V_1 decreases below $-E$, I_3 "jumps" to the lower segment. The trajectory then remains on the lower segment until V_1 exceeds E, when it "jumps" back to the upper segment. This behavior becomes apparent when I_3 is plotted against V_1, as shown in Fig. 16.



Figure 16. Simulation of chaotic trajectory in Saito's oscillator showing how the fast dynamics associated with I_3 cause the trajectory to be confined to the outer portions of the V–I characteristic N_R and to produce "jumps" between these segments.



Figure 17. Simulation of Saito's oscillator.

Table 2. Component List for the Practical Implementation of Saito's Oscillator Shown in Fig. 18

Element	Description	Value
A_1	Op amp (1/2 AD712 or equivalent)	
A_2	Op amp (1/2 AD712 or equivalent)	
C	Capacitor	4.7 nF
R_1	Potentiometer	5 kΩ
R_2	1/4 W resistor	1 kΩ
R_3	1/4 W resistor	1 kΩ
R_4	1/4 W resistor	3.3 kΩ
R_5	1/4 W resistor	10 kΩ
R_6	1/4 W resistor	10 kΩ
R_7	1/4 W resistor	100 Ω
D_1	Zener diode	2.7 V
D_2	Zener diode	2.7 V
L	Inductor (TOKO type 10RB)	100 mH

Chaos Generation Mechanism in Saito's Oscillator

In Saito's oscillator, stretching is accomplished by the negative resistor R which adds energy to the circuit to separate trajectories. The "hysteresis" element switches the trajectory between two two-dimensional regions to keep it bounded. Figure 17 shows a simulation of Saito's circuit with $R = -3$ kΩ, $L = 100$ mH, $C = 4.7$ nF, $L_0 = 1$ nH, $R_a = -3.3$ kΩ, $R_b = 10$ kΩ, and $I = 250$ μA.

Practical Implementation of Saito's Oscillator

A practical implementation of Saito's oscillator is shown in Fig. 18. The negative resistor R is implemented by means of a negative-resistance converter (A_1, R_1, R_2, R_3). Provided that $R_2 = R_3$, then $R = -R_1$. The nonmonotone current-controlled nonlinear "hysteresis" resistor is constructed using a second negative resistance converter. The breakpoint I is chosen by means of zener diodes D_1 and D_2 such that op amp A_1 remains in its linear regime. The saturation voltages at node 5 are given by $E_s \approx 2.7\text{ V} + 0.7\text{ V} = 3.4$ V. A complete list of components is given in Table 2.

SPICE Simulation of Saito's Oscillator

Saito's circuit is characterized by *slow-fast* dynamics: slow two-dimensional dynamics associated with the outer segments of the V–I characteristic of the negative resistor, and fast one-dimensional parasitic dynamics associated with the "jump" through the inner region. Circuits of this type, which are characterized by time scales that differ by several orders of magnitude, are called *stiff* systems. Care must be taken when solving the differential equations to account for the abrupt change in dynamical behavior as the trajectory passes through the inner region (35).

Figure 19 shows a simulation of Saito's circuit using the SPICE deck in Fig. 20.

Chaotic Colpitts Oscillator

Chua's oscillator and Saito's oscillator have been designed with analysis in mind. Their piecewise-linear nature makes analysis and implementation straightforward. In particular, the fast dynamics in Saito's oscillator produces a relatively simple discrete-time equivalent of this system.



Figure 18. Practical implementation of Saito's oscillator using an op amp, resistors, and zener diodes to implement the current-controlled nonlinear resistor.



Figure 19. SPICE simulation of Saito's oscillator.

```
SAITO'S OSCILLATOR

L    1  4    100M
C    4  0    4.7N
* NEGATIVE RESISTOR (VNIC)
V+  111  0    DC 9
V-   0   222  DC 9
XA1   3 1 111 222 2 AD712
R1   3  0    3.0K
R2   2  3    1.0K
R3   1  2    1.0K
* HYSTERESIS ELEMENT (INIC)
XA2   6   4 111 222 8 AD712
R4   6  0    3.3K
R5   5  6    10K
R6   4  5    10K
R7   8  5    100
D1   5  7    ZENER2E7
D2   0  7    ZENER2E7

* 2.7V ZENER DIODE
.MODEL ZENER2E7 D(BV=2.7)

* AD712 SPICE Macro-model   1/91, Rev. A
* Copyright 1991 by Analog Devices, Inc.
* (reproduced with permission)
*
.SUBCKT AD712 13 15 12 16 14
*
VOS 15 8 DC 0
EC 9 0 14 0 1
C1 6 7 .5P
RP 16 12 12K
GB 11 0 3 0 1.67K
RD1 6 16 16K
RD2 7 16 16K
ISS 12 1 DC 100U
CCI 3 11 150P
GCM 0 3  0 1  1.76N
GA 3 0  7 6  2.3M
RE 1 0 2.5MEG
RGM 3 0 1.69K
VC 12 2 DC 2.8
VE 10 16 DC 2.8
RO1 11 14 25
CE 1 0 2P
RO2 0 11 30
RS1 1 4 5.77K
RS2 1 5 5.77K
J1 6 13 4 FET
J2 7 8 5 FET
DC 14 2 DIODE
DE 10 14 DIODE
DP 16 12 DIODE
D1 9 11 DIODE
D2 11 9 DIODE
IOS 15 13 5E-12
.MODEL DIODE D
.MODEL FET PJF(VTO=-1 BETA=1M IS=25E-12)
.ENDS

.IC V(1)=1M V(4)=1M
.TRAN 0.1MS 15MS 5MS
.OPTIONS RELTOL=1.0E-5 ABSTOL=1.0E-5
.END
```

Figure 20. SPICE deck to simulate the transient response of Saito's oscillator. Node numbers are as in Fig. 18. The op amps are modeled by the Analog Devices AD712 macromodel.



Figure 21. Chaotic Colpitts oscillator.

Most electronic oscillators are not piecewise-linear, and the active elements are as likely to be transistors as negative-resistance converters. Provided that the circuits satisfy the necessary conditions for chaos, it is possible that they will exhibit complex steady-state behavior.

A drawback of both Chua's circuit and Saito's circuit is that they are limited to relatively low frequency operation because of the requirements that the nonlinear element should be resistive and piecewise-linear. Novel applications of chaos are now driving the demand for high-frequency chaotic circuits derived from conventional oscillator topologies.

Recall that a harmonic oscillator is usually designed to have a linearized loop gain of unity and a soft nonlinearity to bound the amplitude of the oscillation. By increasing the loop gain beyond unity and employing a hard nonlinearity, chaos can be produced.

The Colpitts oscillator shown in Fig. 21 consists of a linear inductor L with series resistance R_L, a bipolar junction transistor Q, a linear resistor R_{EE}, and two linear capacitors C_1 and C_2.

Assuming that the transistor acts as a purely resistive element, this oscillator can be described by a system of three autonomous state equations:

$$C_1 \frac{dV_{CE}}{dt} = I_L - I_C$$
$$C_2 \frac{dV_{BE}}{dt} = -\frac{V_{EE} + V_{BE}}{R_{EE}} - I_L - I_B$$
$$L \frac{dI_L}{dt} = V_{CC} - V_{CE} + V_{BE} - I_L R_L$$

where, in common-emitter configuration, I_C is written as a function of V_{BE} and V_{CE}.

Table 3. Component List for the Practical Implementation of the Colpitts Oscillator Shown in Fig. 21

Element	Description	Value
R_L	Potentiometer	50 Ω
L	Inductor	100 μH
Q	NPN bipolar transistor	2N2222A
C_1	Capacitor	47 nF
C_2	Capacitor	47 nF
R_{EE}	1/4 W resistor	400 kΩ

When the loop gain is slightly greater than unity and the quality factor of the resonant circuit is high, the transistor in the oscillator remains in its forward active region of operation, and the voltage waveform V_{CE} is almost sinusoidal. By making the loop gain greater than unity and reducing the quality factor, this circuit can exhibit a variety of complex behaviors, including chaos (16,17).

Chaos Generation Mechanism in the Chaotic Colpitts Oscillator

By selecting a sufficiently large small-signal loop gain, the oscillation V_{CE} grows rapidly, the transistor switches off, V_{BE} is driven negative, and then increases slowly until the transistor switches on again. Stretching results from the high gain of the transistor in its forward active region; folding is caused by the spiral decay in the cutoff region.

Practical Implementation of the Chaotic Colpitts Oscillator

A list of components for the chaotic Colpitts oscillator shown in Fig. 21 is given in Table 3. This oscillator exhibits a series of period-doubling bifurcations as R is varied from 0 to 50 Ω. Figure 22 shows a simulation of the chaotic Colpitts oscillator using the SPICE deck in Fig. 23.



Figure 22. SPICE simulation of the Colpitts oscillator in Fig. 21 with V_{CC} = 5 V, V_{EE} = −5 V, R_L = 33 Ω, L = 100 μH, C_1 = 47 μF, C_2 = 47 μF, and R_{EE} = 400 Ω. Q is a type 2N2222A transistor. Vertical axis: V_{BE}; horizontal axis: V_{CE}.

NONAUTONOMOUS CHAOTIC CIRCUITS

Thus far we have considered only autonomous systems where no external forcing signal is applied. An important class of circuits that may exhibit chaos includes those that are driven by a periodic signal. Because the vector field is time-varying in this case, these circuits are called *nonautonomous*. While at least three energy-storage elements are necessary to produce chaos in an autonomous oscillator, chaos can occur in a second-order circuit that is subject to periodic forcing.

Forced Neon Bulb Relaxation Oscillator

One of the earliest recorded observations of chaos in an electronic circuit is the driven neon bulb relaxation oscillator studied by van der Pol and van der Mark in 1927 (4,5). The circuit, shown in Fig. 24, consists of a high voltage dc source E attached via a large series resistance R to a neon bulb and capacitor C_1, which are connected in parallel; this forms the basic relaxation oscillator. Initially the capacitor is discharged and the neon bulb is nonconducting. The dc source charges C_1 with a time constant RC until the voltage across the neon bulb is sufficient to turn it on. Once lit, the bulb presents a shunt low-resistance path to the capacitor. The voltage across the capacitor falls exponentially until the neon arc is quenched, the bulb is returned to its off state, and the cycle repeats.

As the capacitance C_1 is increased smoothly, the circuit exhibits jumps from one (periodic) mode-locked state to another. For a critical value of the amplitude of the driving signal, the pattern of mode-lockings has a self-similar fractal structure consisting of an infinite number of steps. This is called a Devil's staircase (36).

When the amplitude of the forcing signal is greater than the critical value, the steps of the staircase overlap. Van der Pol noted that "often an irregular noise is heard in the telephone receiver before the frequency jumps to the next lower value"; this is chaos.

The frequency-locking behavior of the driven neon bulb oscillator circuit is characteristic of forced oscillators that contain two competing frequencies: the natural frequency f_0 of the undriven oscillator and the driving frequency f_s. If the amplitude of the forcing is small, either *quasi-periodicity* or *mode-locking* occurs. For a sufficiently large amplitude of the forcing, the system may exhibit chaos.

Figure 25 shows experimentally observed mode locking in a driven neon bulb oscillator. Magnifications of the staircase are shown in Fig. 26. For driving signals with amplitudes greater than that shown, the monotonicity of the staircase is lost and chaos occurs.

The presence of a single dynamic element (the capacitor) in Fig. 24 might suggest that this is a first-order system, but a first-order circuit with periodic forcing cannot exhibit chaos. The "hidden" second state is associated with the fast transit dynamics of the neon bulb. The neon bulb may be modeled as a nonmonotone current-controlled nonlinear resistor with a parasitic series inductor (5).

Driven *RL*-Diode Circuit

One of the simplest nonautonomous chaotic circuits is the series connection of a linear resistor, a linear inductor, and a

```
COLPITTS OSCILLATOR

VCC 1 4 PWL(0 0 1N 5 5M 5)
VEE 5 4 DC -5
RL  1 2 33
L   2 3 100U
Q   3 4 0 Q2N2222A
C1  3 0 47N
C2  4 0 47N
REE 0 5 400

.MODEL Q2N2222A NPN(IS=14.34F XTI=3 EG=1.11 VAF=74.03 BF=255.9 NE=1.307
+        ISE=14.34F IKF=.2847 XTB=1.5 BR=6.092 NC=2 ISC=0 IKR=0 RC=1
+        CJC=7.306P MJC=.3416 VJC=.75 FC=.5 CJE=22.01P MJE=.377 VJE=.75
+        TR=46.91N TF=411.1P ITF=.6 VTF=1.7 XTF=3 RB=10)

.OPTIONS RELTOL=1E-5 ABSTOL=1E-5
.TRAN 10N 4M 3M
.END
```

Figure 23. SPICE deck to simulate the transient response of the Colpitts oscillator shown in Fig. 21.

pn-junction, as shown in Fig. 27, which can exhibit chaotic behavior when driven by a sinusoidal voltage source (7,8). In this case chaos is due to parasitic nonlinear capacitive effects in the diode. The behavior of the circuit can be confirmed by using SPICE (9) (see Figs. 28 and 29).

Driven Negative-Resistance Circuit

Chaos in the driven *RL*-diode circuit is due to relatively complex nonlinear dynamical behavior in the *pn*-junction diode. A simpler nonautonomous circuit containing only *linear* energy-storage elements is the driven negative-resistance circuit shown in Fig. 30. This consists of a series connection of a periodic voltage source, a linear resistor, a linear inductor, and a parallel connection of a nonmonotone voltage-controlled nonlinear resistor and a linear capacitor.

This circuit is described by a pair of first-order nonautonomous ordinary differential equations:

$$\frac{dV_1}{dt} = -\frac{1}{C_1} f(V_1) + \frac{1}{C_1} I_2$$
$$\frac{dI_2}{dt} = -\frac{1}{L} V_1 - \frac{R}{L} I_2 + \frac{A}{L} \sin(2\pi f_s t)$$

where the voltage-controlled nonlinear resistor is described by $I_R = f(V_R)$.



Figure 24. Driven neon bulb relaxation oscillator.

The case of a cubic nonlinearity—the electrical analog of the forced Duffing equation—has been studied extensively (6). The undriven system has three equilibrium points, one of which is a saddle. The two remaining equilibria are stable fixed points. Chaos arises when the trajectory is driven close to the saddle.

The same qualitative behavior, shown in Fig. 31, occurs when a simpler piecewise-linear nonlinearity is used instead of a cubic. Here $A = 2$ V, $f_s = 5000$ Hz, $R = 660\ \Omega$, $L = 33$ mH, and $C = 68$ nF (37).



Figure 25. Experimentally measured staircase structure of lockings for a forced neon bulb relaxation oscillator. The winding number is given by f_s/f_d, the ratio of the frequency of the sinusoidal driving signal to the average frequency of current pulses through the bulb.



Figure 26. Magnification of Fig. 25 showing self-similarity.

The V–I characteristic of the nonlinear resistor is as shown in Fig. 32. The relationship may be written explicitly as

$$f(V_R) = G_b V_R + \tfrac{1}{2}(G_a - G_b)(|V_R + E| - |V_R - E|)$$

where $G_a = -2.2$ mS, $G_b = 1$ mS, and $E = 1.6875$ V. This element is readily implemented by means of a negative-resistance converter.

A complete circuit realization of the driven negative-resistance oscillator with a piecewise-linear nonlinear resistor is shown in Fig. 33. A component list for the practical implementation of this circuit is given in Table 4. The behavior of the circuit may be confirmed by SPICE simulation (see Figs. 34 and 35).

DISCRETE-TIME CHAOTIC CIRCUITS

Although a discrete-time, *discrete-state* deterministic dynamical system may exhibit long periodic steady-state trajectories, it cannot exhibit chaos. By contrast, a discrete-time system of order one or more can exhibit chaos if it has continuous state variables and is described by a nonlinear map. If the system



Figure 27. Driven RL-diode circuit.

```
DRIVEN RL-DIODE CIRCUIT

D    3  0    DIODE
R    1  2    15
L    2  3    10.0M
VS   1  0    SIN(0 6 100K)

.MODEL DIODE D(IS=8.3FA RS=9.6 TT=4US CJ0=300PF M=0.4 VJ=0.75)

.TRAN 0.001US 2MS 1MS
.OPTIONS RELTOL=1.0E-5 ABSTOL=1.0E-5
.END
```

Figure 28. SPICE deck to simulate the behavior of the RL-diode circuit shown in Fig. 27.



Figure 29. Spice simulation of driven RL-diode circuit.



Figure 30. Driven negative-resistance oscillator.



Figure 31. Simulation of driven negative-resistance circuit with piecewise-linear resistor as in Fig. 32.



Figure 32. V–I characteristic of the negative resistor in Fig. 30.



Figure 33. Practical implementation of driven negative-resistance oscillator using an op amp, resistors, and zener diodes to implement the voltage-controlled nonlinear resistor.

Table 4. Component List for the Practical Implementation of Negative-Resistance Circuit Shown in Fig. 33

Element	Description	Value
A_1	Op amp (1/2 AD712 or equivalent)	
R	Potentiometer	1 kΩ
L	Inductor	33 mH
C	Capacitor	68 nF
R_1	1/4 W resistor	1 kΩ
R_2	1/4 W resistor	2.2 kΩ
R_3	1/4 W resistor	1 kΩ
R_4	1/4 W resistor	100 Ω
D_1	Zener diode	4.7 V
D_2	Zener diode	4.7 V

is first-order, then the nonlinear map must also be noninvertible.

Switched-Capacitor Chaotic Circuit

A continuous-state, discrete-time dynamical system of the form

$$\mathbf{X}_{k+1} = \mathbf{G}(\mathbf{X}_k)$$

can be implemented electronically using switched-capacitor (SC) circuits. Such circuits may exhibit chaos if the map $\mathbf{G}$ is nonlinear and at least one of the eigenvalues of $\mathbf{D_x G}(\cdot)$ has modulus greater than unity in magnitude for some states $\mathbf{X}$.

One of the most widely used deterministic "random" number generators is the *linear congruential generator,* which is a discrete-time dynamical system of the form

$$X_{k+1} = (AX_k + B) \bmod M, \qquad k = 0, 1, \ldots \tag{1}$$

where A, B, and M are called the *multiplier, increment,* and *modulus,* respectively.

If $A > 1$, then all equilibrium points of (1) are unstable. With the appropriate choice of constants, this system exhibits a chaotic solution with a positive Lyapunov exponent equal to $\ln A$. However, if the state space is discrete, for example, in the case of digital implementations of (1), then every steady-state orbit is periodic with a maximum period equal to the number of distinct states in the state space; such orbits are termed *pseudorandom.*

By using an analog state space, a truly chaotic sequence can be generated. A discrete-time chaotic circuit with an analog state space may be realized in switched-capacitor technology.

Example: Parabolic Map. Figure 36 shows an SC realization of the parabolic map

$$x_{k+1} = V - 0.5x_k^2$$

which, by the change of variables $X_k = Ax_k + B$, with $B = 0.5$ and $A = (-1 + \sqrt{1 + 2\,V})/(4\,V)$, and $0\text{ V} \le V \le 4\text{ V}$, is equivalent to the logistic map

$$X_{k+1} = PX_k(1 - X_k)$$

with $P = 1/(2A)$ in the range $2 \le P \le 4$ (19).

```
DRIVEN NEGATIVE-RESISTANCE OSCILLATOR

VS   1  0    SIN(0 2.0 5K)
R    1  2    660
L    2  3    33.0M
C    3  0    68.0N
* VOLTAGE-CONTROLLED NONLINEAR RESISTOR
V+  111  0   DC 9
V-   0  222  DC 9
R1   5  0    1K
R2   4  5    2.2K
R3   3  4    1K
R4   7  4    100
XA1  3 5 111 222 7 AD712
D1   4  6    ZENER4E7
D2   0  6    ZENER4E7
.MODEL ZENER4E7 D(BV=4.7)

* AD712 SPICE Macro-model   1/91, Rev. A
* Copyright 1991 by Analog Devices, Inc.
* (reproduced with permission)
*
.SUBCKT AD712 13 15 12 16 14
*
VOS 15 8 DC 0
EC 9 0 14 0 1
C1 6 7 .5P
RP 16 12 12K
GB 11 0 3 0 1.67K
RD1 6 16 16K
RD2 7 16 16K
ISS 12 1 DC 100U
CCI 3 11 150P
GCM 0 3  0 1  1.76N
GA 3 0  7 6  2.3M
RE 1 0 2.5MEG
RGM 3 0 1.69K
VC 12 2 DC 2.8
VE 10 16 DC 2.8
RO1 11 14 25
CE 1 0 2P
RO2 0 11 30
RS1 1 4 5.77K
RS2 1 5 5.77K
J1 6 13 4 FET
J2 7 8 5 FET
DC 14 2 DIODE
DE 10 14 DIODE
DP 16 12 DIODE
D1 9 11 DIODE
D2 11 9 DIODE
IOS 15 13 5E-12
.MODEL DIODE D
.MODEL FET PJF(VTO=-1 BETA=1M IS=25E-12)
.ENDS

.TRAN 0.1MS 60MS 10MS
.OPTIONS RELTOL=1.0E-5 ABSTOL=1.0E-5
.PRINT TRAN V(1) V(3)
.END
```

Figure 34. SPICE deck to simulate the transient response of the driven negative-resistance oscillator. Node numbers are as in Fig. 33. The op amps are modeled by the Analog Devices AD712 macromodel.



Figure 35. SPICE simulation of driven negative-resistance circuit.

In the case considered, $V = P(P - 2)/2$. For $0 \text{ V} \leq V < 1.5$ V, $2 \leq P < 3$ and the steady-state solution of the SC parabolic map is a fixed point. As the bifurcation parameter V is increased from 1.5 to 3 V, the circuit undergoes a series of period-doubling bifurcations to chaos. $V = 4$ V corresponds to fully-developed chaos on the open interval $(0 < X_k < 1)$ in the logistic map with $P = 4$.

CONCLUDING REMARKS

We have illustrated a very limited selection of autonomous and nonautonomous electronic circuits that exhibit chaos. So many other electronic circuits and systems are now known to exhibit complex nonlinear dynamical behavior, including chaos, that it would be impossible to mention all of them. The interested reader is referred to special issues of the *IEEE Transactions on Circuits and Systems* (October 1993) and the



Figure 36. Switched-capacitor realization of the parabolic map $x_{k+1} = V - 0.5X_k^2$. The switches labeled o and e are driven by the odd and even phases, respectively, of a nonoverlapping two-phase clock.

Philosophical Transactions of the Royal Society London A (October 1995) on "Chaos in nonlinear electronic circuits" and "Chaotic behavior in electronic circuits," respectively.

Low-dimensional chaos is now well understood and is finding applications in noise (dither) generation, targeting, and wideband communications. Current research into chaos in electronic circuits is aimed at developing hyperchaotic circuits (38), robust high-frequency chaos generators, and circuit techniques for producing or suppressing chaos.

BIBLIOGRAPHY

1. D. Ruelle and F. Takens, On the nature of turbulence, *Commun. Math. Phys.,* **20**: 167–192, 1971.
2. J. M. T. Thompson and H. B. Stewart, *Nonlinear Dynamics and Chaos,* New York: Wiley, 1986.
3. E. Ott, *Chaos in Dynamical Systems,* Cambridge: Cambridge Univ. Press, 1993.
4. B. van der Pol and J. van der Mark, Frequency demultiplication, *Nature,* **120**: 363–364, 1927.
5. M. P. Kennedy and L. O. Chua, Van der Pol and chaos, *IEEE Trans. Circuits Syst.,* **33**: 974–980, 1986.
6. Y. Ueda and N. Akamatsu, Chaotically transitional phenomena in the forced negative-resistance oscillator, *IEEE Trans. Circuits Syst.,* **28**: 217–224, 1980.
7. P. Linsay, Period doubling and chaotic behavior in a driven anharmonic oscillator, *Phys. Rev. Lett.,* **47**: 1349–1392, 1981.
8. J. Testa, J. Pérez, and C. Jeffries, Evidence for universal chaotic behavior of a driven nonlinear oscillator, *Phys. Rev. Lett.,* **48**: 714–717, 1982.
9. A. Azzouz, R. Duhr, and M. Hasler, Transition to chaos in a simple nonlinear circuit driven by a sinusoidal source, *IEEE Trans. Circuits Syst.,* **30**: 913–914, 1983.
10. L. O. Chua, Global unfolding of Chua's circuit. *IEICE Trans. Fundam. (Special Issue: Chaos, Neural Netw., Numerics),* **E76-A**: 704–734, 1993.
11. R. N. Madan, (ed.), *Chua's Circuit: A Paradigm for Chaos,* Singapore: World Scientific, 1993.
12. M. P. Kennedy, Three steps to chaos. Part II: A Chua's circuit primer, *IEEE Trans. Circuits Syst. (Special Issue: Chaos Nonlinear Electron Circuits, Part A: Tutorial Rev.),* **40**: 657–674, 1993.
13. R. W. Newcomb and N. El-Leithy, A binary hysteresis chaos generator, *Proc. ISCAS '84,* Montreal, Canada, 1984, pp. 856–859.
14. T. Saito, On a hysteresis chaos generator, *Proc. ISCAS '85,* Kyoto, 1985, pp. 847–849.
15. T. Saito and S. Nakagawa, Chaos from a hysteresis and switched circuit, *Philos. Trans. R. Soc. London,* **353**: 47–57, 1995.
16. M. P. Kennedy, Chaos in the Colpitts oscillator, *IEEE Trans. Circuits Syst. I, Fundam. Theory Appl.,* **41**: 771–774, 1994.
17. M. P. Kennedy, On the relationship between the chaotic Colpitts oscillator and Chua's oscillator, *IEEE Trans. Circuits Syst. I, Fundam. Theory Appl.,* **42**: 376–379, 1995.
18. T. Endo, A review of chaos and nonlinear dynamics in phase-locked loops, *J. Franklin Inst.,* **331B**: 859–902, 1994.
19. A. Rodríguez-Vázquez et al., Chaos from switched-capacitor circuits: Discrete maps, *Proc. IEEE,* **75**: 1090–1106, 1987.
20. D. C. Hamill and D. J. Jefferies, Subharmonics and chaos in a controlled switched-mode power converter, *IEEE Trans. Circuits Syst.,* **35**: 1059–1061, 1988.
21. D. C. Hamill, Learning about chaotic circuits with SPICE, *IEEE Trans. Educ.,* **36**: 28–35, 1993.
22. A. C. Davies, Nonlinear oscillations and chaos from digital filter overflow, *Philos. Trans. R. Soc. London,* **353**: 85–99, 1995.
23. R. Schreier, On the use of chaos to reduce idle-channel tones in delta-sigma modulators, *IEEE Trans. Circuits Syst. I, Fundam. Theory Appl.,* **41**: 539–547, 1994.
24. T. S. Parker and L. O. Chua, *Practical Numerical Algorithms for Chaotic Systems,* New York: Springer, 1989.
25. M. P. Kennedy, Basic concepts of nonlinear dynamics and chaos, C. Toumazou (ed.), *Circuits and Systems Tutorials,* London: IEEE Press, 1994, pp. 289–313.
26. L. O. Chua, M. Komuro, and T. Matsumoto, The double scroll family. Parts I and II, *IEEE Trans. Circuits Syst.,* **33**: 1073–1118, 1986.
27. L. O. Chua, C. A. Desoer, and E. S. Kuh, *Linear and Nonlinear Circuits,* New York: McGraw-Hill, 1987.
28. M. P. Kennedy, ABC (Adventures in Bifurcations and Chaos): A program for studying chaos. *J. Franklin Inst. (Special Issue: Chaos Electron. Circuits,* **331B**: 529–556, 1994.
29. M. P. Kennedy, Robust op amp realization of Chua's circuit, *Frequenz,* **46**(3–4): 66–80, 1992.
30. J. M. Cruz and L. O. Chua, A CMOS IC nonlinear resistor for Chua's circuit, *IEEE Trans. Circuits Syst.,* **39**: 985–995, 1992.
31. A. Vladimirescu, *The SPICE Book,* New York: Wiley, 1994.
32. Analog Devices, Inc., *SPICE Model Library,* Norwood, MA, 1991. Release C 1/91.
33. M. Vidyasagar, *Nonlinear Systems Analysis,* Englewood Cliffs, NJ: Prentice-Hall, 1978.
34. M. P. Kennedy and L. O. Chua, Hysteresis in electronic circuits: A circuit theorist's perspective, *Int. J. Circuit Theory Appl.,* **19**: 471–515, 1991.
35. L. O. Chua and P. M. Lin, *Computer-Aided Analysis of Electronic Circuits: Algorithms and Computational Techniques,* Englewood Cliffs, NJ: Prentice-Hall, 1975.
36. M. P. Kennedy, K. R. Krieg, and L. O. Chua, The devil's staircase: The electrical engineer's fractal, *IEEE Trans. Circuits Syst.* **36**: 1133–1139, 1989.
37. J. G. Lacy, A simple piecewise-linear non-autonomous circuit with chaotic behavior, *Int. J. Bifurcation Chaos,* **6**(11): 2097–2100, 1996.
38. T. Matsumoto, L. O. Chua, and K. Kobayashi, Hyperchaos: Laboratory experiment and numerical confirmation, *IEEE Trans. Circuits Syst.,* **33**: 1143–1147, 1986.

MICHAEL PETER KENNEDY
University College Dublin